MACAULAY2 NOTES – MATH 538
FALL 2013

KARL SCHWEDE

1. Wednesday, September 18th, 2013

We'll learn about modules, tensor and Hom.

Macaulay2 can only deal with finitely presented modules (for obvious reasons). The first module you need to care about is a free module. So let's begin by defining a ring (a quotient ring would work fine too).

```
R = QQ[x,y,z]
```

While R is a free rank-1 R-module, Macaulay2 does not treat it as such. To consider a free R-module of rank 5 you can create:

```
M = R^5
```

Of course, many modules are also presented with relations, these are cokernels of module maps. So we begin by making a matrix, a map between free modules.

```
m = matrix{ {x, x*y, z^2}, {z, z*x, y} }
L = coker m
```

This gives us the module with two generators, say `e1, e2` modulo the three relations $\mathtt{e1}*\mathtt{x}+\mathtt{e2}*\mathtt{z}, \mathtt{e1}*\mathtt{x}*\mathtt{y}+\mathtt{e2}*\mathtt{z}*\mathtt{x}, \mathtt{e1}*\mathtt{z}^2+\mathtt{e2}*\mathtt{y}$. Notice how Macaulay2 reminds you of the source and target of the matrix map?

Exercise 1.1. Create a module N with three generators $\mathtt{a}, \mathtt{b}, \mathtt{c}$ modulo the two relations $\mathtt{a}*\mathtt{x}+\mathtt{b}*\mathtt{y}+\mathtt{c}*\mathtt{z}, \mathtt{a}*\mathtt{z}^2+\mathtt{b}*\mathtt{x}^2+\mathtt{c}*\mathtt{y}^2$.

Of course, sometimes we create modules in different ways, especially submodules.

```
I = ideal(x,y,z)
S = I*M
T = I*N
```

We just created submodules of M and N. Macaulay is expressing these two modules as submodules. But perhaps you want to directly find their generators and relations. To do that run

```
presentation S
presentation T
```

These are big presentations. Use `minimalPresentation` (or equivalently `prune`) to cut down on the generators (this may or may not work, depending if it is possible to cut down on the generators).

Exercise 1.2. Create module such that it is possible to `prune` the module and drastically cut down on the number of generators.

Of course, sometimes you just want to view an ideal as a module, or a quotient as a module.

```
O = module I
presentation O
P = R^1/I
presentation P
```

Let's next talk about maps between modules. Say I want to define a map from $\mathtt{M} = \mathtt{R}^5$ to `L`. I need to specify where the generators of `M` go in terms of the generators of `L`.

```
f = map(L, M, matrix{{x,1,y,z*x,5*z}, {x*y,2*x, 1, y*z, z^2}})
isWellDefined f
```

If M wasn't a free module, we'd have to worry whether or not this is well defined.

Exercise 1.3. Create an example of a ring map which is not well defined. Check it with Macaulay2.

We can create Hom and $\otimes$ modules too using `Hom` and `**`.

```
Hom(L, M)
L ** M
```

Exercise 1.4. Find an example of modules A and B such that $Hom(A, B) = 0$ even in the case that B is not a free module and A is nonzero. Verify your work with Macaulay2.

Exercise 1.5. Find two nonzero modules A and B such that $A \otimes B = 0$. Verify your work in Macaulay2.

Department of Mathematics, The Pennsylvania State University, University Park, PA, 16802, USA
E-mail address: `schwede@math.psu.edu`